

Enhancing RTF Output with RTF Control Words and In-Line Formatting

Lori S. Parsons, Ovation Research Group, a division of ICON Clinical Research, Seattle, WA

ABSTRACT

The SAS® Output Delivery System (ODS) allows output to be printed directly to a Rich Text Format (RTF) file. RTF can be read by word processors such as Microsoft Word and WordPerfect. In SAS 8.1, options were added to enhance SAS TITLES and FOOTNOTES. Also added was the ability to use RTF control words within cells. In SAS 8.2, ODS supported in-line formatting and added the TEXT= option. All of these formatting methods give us the ability to insert formatting text into a cell or paragraph and can be used to enhance output. In this paper, an RTF file will be created with enhanced SAS TITLES, footnote symbols in variable labels and values, custom footnotes, an inserted image, and an inserted hyperlink. The code was run with SAS Version 9.1.2 on Windows XP Professional. The RTF files are viewed with Microsoft® Office Word 2003.

INTRODUCTION

Rich Text Format (RTF) is a file type used to transfer formatted text and graphics between applications, even those that run on different platforms such as Windows, IBM, and Macintosh. With RTF, documents created under different operating systems and with different software applications can be transferred between those operating systems and applications.

It has become increasingly popular to produce output in SAS and then distribute it to others who will view the file in a word processor. Being able to customize SAS output through RTF code is an invaluable tool. Customized reports can be produced without having to manually edit the file before distribution.

In the example that follows, an RTF file will be created that contains the following: An image, a URL link, formatted SAS titles, customized text, and footnote symbols in a SAS title, variable label, and formatted variable value. The pertinent SAS code has been provided in the body of the paper. The complete SAS code can be found in Appendix I.

BACKGROUND

RTF Syntax

An RTF file consists of unformatted text, control words, control symbols, and groups. A *control word* is a specially formatted command that RTF uses to mark printer control codes and information that applications use to manage documents. A control word cannot be longer than 32 characters. A control word takes the following form: \LetterSequence<Delimiter>. A backslash begins each control word. The LetterSequence is made up of lowercase alphabetic characters (a-z). RTF is case sensitive.

A *control symbol* consists of a backslash followed by a single, nonalphabetic character. For example, \~ represents a nonbreaking space. Control symbols take no delimiters.

A *group* consists of text and control words or control symbols enclosed in braces ({}). The opening brace ({) indicates the start of the group and the closing brace (}) indicates the end of the group. Each group specifies the text affected by the group and the different attributes of that text. The RTF file can include groups for fonts, styles, screen color, pictures, footnotes, comments (annotations), headers and footers, summary information, fields, and bookmarks, as well as document-, section-, paragraph-, and character-formatting properties. Formatting specified within a group affects only the text within that group.

The control properties of certain control words (such as bold, italic, keep together, and so on) have only two states. When such a control word has no parameter or has a nonzero parameter, it is assumed that the control word turns on the property. When such a control word has a parameter of 0, it is assumed that the control word turns off the property. For example, \b turns on bold, whereas \b0 turns off bold.

Footnote Symbol Standards

The explanatory footnote symbols used in the presented example will follow the guidelines presented by the *American Medical Association Manual of Style*. Footnotes will be indicated by the following symbols, used in the order shown: * (asterisk), † (dagger), ‡ (double dagger), § (section marker), || (parallels), ¶ (paragraph symbol), and # (number sign). Per the guidelines, these symbols are repeated, doubled, if more are needed.

Specifying Special Characters

Special characters cannot be typed directly into a text string. To display these special characters, a hexadecimal value can be used. Hexadecimal values are any two-digit hexadecimal numbers enclosed in quotation marks and followed by the letter x; for example, '3D'x. In double-byte character sets, the hexadecimal values contain four digits; for example, '4E60'x.. To display characters with hexadecimal values, specify the font that contains the special character and place the hexadecimal value in the text string. The following TITLE statements use hexadecimal 86 to produce † and hexadecimal 87 to produce ‡ in the Arial font.

```
title1 font=arial '86'x 'Text Goes Here'; /* print a dagger before text */
title2 font=arial '87'x 'Text Goes Here'; /* print a double dagger */
```

When creating an RTF file in SAS, if the symbol already exists in a document (a WORD document, for example), it can simply be “cut-and-paste” into the appropriate SAS code. The symbol will then appear in the RTF file that is created by SAS. For example, the symbols for the dagger and double dagger have been pasted into the following TITLE statements.

```
title1 font=arial '† Text Goes Here '; /* print a dagger before text */
title2 font=arial '‡ Text Goes Here '; /* print a double dagger before text */
```

METHODS

Creating an RTF file in SAS

The ODS code to create a basic RTF file is very simple. The code is:

```
ODS RTF file = "the path name goes here/filename.RTF";
/* SAS procedures go here */
run;
quit;
ODS RTF close;
```

ODS RTF FILE = opens the destination and file; ODS RTF CLOSE closes the destination and file. The RUN and QUIT commands before the ODS RTF close statement are required. If QUIT is omitted, the output from the last procedure may not be included in the document.

SAS TITLE and FOOTNOTE Options

Beginning with SAS version 8.1, the style of titles and footnotes can be modified by applying any of eight options on the TITLE or FOOTNOTE statement. The eight options are listed in **Table 1**. This is probably the easiest and most straight-forward way to modify a title or footnote.

Table 1: SAS TITLE and FOOTNOTE Statement Options

SAS Option	Function
BOLD	Bolds the title or footnote.
COLOR= <i>color</i>	Supplies the color of the title or footnote.
BCOLOR= <i>color</i>	Supplies the background color of the title or footnote.
FONT= <i>font</i>	Supplies the font for the title or footnote.
HEIGHT= <i>text-height</i> < <i>units</i> >	Supplies the size for the title or footnote.
JUSTIFY= LEFT RIGHT CENTER	Left-justifies, right-justifies, or centers the title or footnote.
LINK= ' <i>url</i> '	Specifies the hyperlink to use for the title or footnote.
UNDERLIN= 0...3	Specifies an underline. Values of 1, 2 and 3 underline with an increasingly thicker line. UNDERLIN=0 halts underlining for subsequent text.

The following code uses all eight options to produce the results in **Figure 1**:

```

title;
footnote;
options orientation=portrait nodate center pageno=1;

ods rtf file = "H:\Admin\SGF07\Output\SGF07_Ex1.rtf" startpage=no;
/* Define SAS Titles */
/* The title of the first table will be contained in title5 */
title1 j=1 font=elephant height = 30 pt color=dab bcolor=vpab
      "Einstein Middle School" ;
title2 j=1 height = 14 pt link="http://www.schools.org/einstein/"
      "http://www.schools.org/einstein/";
title4 j=c font=arial height = 18 pt color=mob bold underlin = 1
      "Summary of Class Enrollment for 2006/2007";
title6 j=c font=arial height = 14 pt color=bipb bold
      "Gender of Enrolled Students";
proc freq data = sashelp.class;
  table sex;
run;
quit;
ods rtf close;

```

Figure 1: RTF Document After Using SAS TITLE and FOOTNOTE Statement Options

Einstein Middle School				
http://www.schools.org/einstein/				
Summary of Class Enrollment for 2006/2007				
Gender of Enrolled Students				
<i>The FREQ Procedure</i>				
Sex	Frequency	Percent	Cumulative Frequency	Cumulative Percent
F	9	47.37	9	47.37
M	10	52.63	19	100.00

In-Line Formatting

Beginning in SAS 8.2, ODS supports the ability to insert simple formatting text into a cell or paragraph. This method is probably the second most straight-forward method to customize output. With this you can further customize SAS titles and footnotes, add superscripts and subscripts within a cell, insert destination-specific raw text, and add images with PREIMAGE= and POSTIMAGE=. The image can be your company's logo, for example.

To use in-line formatting, you first define an escape character with the ODS statement: ODS ESCAPECHAR = 'escape-character';. Choose an escape character that is not used in code for any other reason. The '^' character should usually be a safe choice and will be used in the examples that follow. **Table 2** lists some ways in-line formatting with the escape character can be used.

Table 2: Sample 8.2 In-line Formatting Code

SAS Option	Function
^S={style-attributes}	Modifies the style of the current paragraph or cell. The syntax is just like that of the style={} syntax in table templates and PROCs REPORT and TABULATE except that no style element is supported to the left of the {}s.
^S={}	A specialized case of the above; reverts the style to the style of the paragraph.
^{super text}	Put text into a superscript.
^{sub text}	Put text into a subscript.
^{dagger}	Print the "dagger" character.
^n	Inserts a new line.
^w	Marks a good place for an optional line-break. Doesn't force a new line, but "suggests" it as a good place if the line must be wrapped.
^_	Inserts a non-breaking space.

The following is the in-line formatting code added to produce the output in **Figure 2**. In this example, an image is added to SAS TITLE1 by using the ^S={} code and PREIMAGE= option.

```
ods escapechar = '^';

title1 j=l font=elephant height = 30 pt color=dab bcolor=vpab
      "^S={preimage='H:\Admin\SGF07\Logo\Einstein.jpg'} Einstein Middle School";
```

A footnote symbol is added to the SAS TITLE4 using the ^{super text} code.

```
title4 j=c font=arial height = 18 pt color=mob bold underlin = 1
      "Summary of Class Enrollment for 2006/2007 ^{super *}";
```

A footnote symbol is added to a formatted variable value using the ^{super text} code.

```
ods select OneWayFreqs;
ods output OneWayFreqs=OneWayFreqs;
proc freq data = sashelp.class;
  table sex;
run;

proc format;
  value $fsex
    'F' = "Female "
    'M' = "Male ^{super †}";
run;
```

Paper 151-2007

```

proc print data = onewayfreqs noobs;
  var sex /
    style(header)=table{font=("Arial, Helvetica",12pt) just=c asis=on
      cellwidth=1.45in}
    style(data)=table{font=("Arial, Helvetica",12pt) just=l asis=on
      cellwidth=1.45in};
  var frequency percent /
    style(header)=table{font=("Arial, Helvetica",12pt) just=c asis=on
      cellwidth=1.45in}
    style(data)=table{font=("Arial, Helvetica",12pt) just=c asis=on
      cellwidth=1.45in};
  format sex $fsex.;
run;

```


A footnote symbol is added to a variable label using the `^`{super text} code.

```

ods select summary;
ods output summary=summary;
proc means data = sashelp.class maxdec=1;
  var age;
run;
proc print data=summary noobs label;
  var age_mean age_min age_max /
    style(header)=table{font=("Arial, Helvetica",12pt) just=c asis=on
      cellwidth=1.45in}
    style(data)=table{font=("Arial, Helvetica",12pt) just=l asis=on
      cellwidth=1.45in};
  label age_mean = "Mean Age ^{super †}";
  label age_min = "Minimum Age";
  label age_max = "Maximum Age";
run;

```

Figure 2: RTF Document After Adding In-line Formatting Code



Using RTF Control Words in SAS

Beginning with SAS version 8.1, RTF control words can be used within cells. Of the three methods presented in this paper, using RTF control words is probably the trickiest, but richest option for controlling the format of output cells.

Table 3 contains a small list of commonly used RTF control words. There are many sources for complete lists of RTF control words. One good web source is from the Microsoft Corporation titled *Rich Text Format (RTF) Specification, Version 1.6*.

Table 3: Sample of RTF Control Words

RTF Control Word	Function
\i	Italicize
\ul	Underline
\dul	Double underline
\line	New line
\bullet	Bullet
\embo	Emboss
\impr	Engrave
\sub	Subscript
\super	Superscript
\outl	Outline
\shad	Shadow
\strike	Strike
\strikedl	Double strike
\uld	Dotted underline
\ulw	Wave underline
\ulth	Thick underline
\cfn	Foreground color
\fs	Font size in half points Usage: \fs24
\highlightN	Highlight
\b	Bold
\ql	Left aligned
\qr	Right aligned
\qc	Center

Custom Text with ODS TEXT= Statement

The ODS TEXT= statement allows text to be added between procedures. Since only one set of titles and footnotes are written out to a file when the STARTPAGE option is set to NO,OFF, or NEVER, this option is extremely useful for describing table information. It is also extremely useful in adding custom footnotes. The ODS TEXT= statement can contain a combination of in-line formatting code and RTF control words.

By default the text is left justified in the PRINTER (PCL, PDF, PS) and RTF files. In an RTF file, the ODS TEXT= statement does not adhere to any in-line style specifications or PROC TEMPLATE instructions for justification. In SAS versions prior to 9.1, RTF code was necessary to force center justification. The following code will create a cell whose right boundary is well past the center of the page (\cellx11520) and center the text in the cell (\qc).

```
ods rtf text="{\trowd\trqc\cellx11520\intbl\qc{Text Goes Here\cell}}{\row}\pard}";
```

Paper 151-2007

In SAS 9.1, the following will work to center justify text:

```
ods rtf text="^S={outputwidth=100% just=c} Text Goes Here";
```


Figure 3 shows the final output using all the formatting methods described in this paper. A custom title for the second table ("Age of Enrolled Students") is created with the following ODS RTF TEXT= statement:

```
/* Code for version 9.1 and higher */
ods rtf text = "^S={outputwidth=100% just=c} {\line\line\b\qc\f3\fs28\cf20
  Age of Enrolled Students}";
```

Custom footnotes were added using the ODS RTF TEXT= statement, in-line formatting code with the escape character, and RTF control words to define formatting.

```
ods rtf text = "^S={outputwidth=100% just=l}
  {\line\line\line _____ \line}";
ods rtf text = "^S={outputwidth=100% just=l}{\f3\fs34\super *}
  {\f3\fs24 Includes summer session.}";
ods rtf text = "^S={outputwidth=100% just=l}{\f3\fs34\super †}
  {\f3\fs24 Two males enrolled in summer session.}";
ods rtf text = "^S={outputwidth=100% just=l}{\f3\fs34\super ‡}
  {\f3\fs24 Age on first day of class.}";
```

Figure 3: RTF Document After Adding Custom Text and RTF Control Words



Paper 151-2007

CONCLUSIONS

It is now possible to enhance RTF files with custom text, titles, and footnotes. Hyperlinks, pictures, and color can also be added. The example presented in this paper uses a combination of SAS options, ODS code, in-line formatting, and RTF control words to create a customized report. This is just a small sample of the customization that can be accomplished.

REFERENCES

American Medical Association Manual of Style: A Guide for Authors and Editors, 9th edition (Edited by: M. Pugh). Chicago, Lippincott Williams & Wilkins, 1998, p 62.

Haworth, Lauren E., *Output Delivery System: The Basics*, Cary, NC: SAS Institute Inc., 2001.

Microsoft Corporation, *Rich Text Format (RTF) Specification, Version 1.6*, <http://msdn.microsoft.com/library/default.asp?url=/library/en-us/dnrtf/spec/html/rtf/spec.asp>, May 1999.

SAS Institute Inc., *In-Line Formatting for ODS in SAS 8.2*, <http://support.sas.com/rnd/base/topics/expv8/inline82.html>.

SAS Institute Inc., *ODS FAQ and Concepts: The RTF Destination*, http://support.sas.com/rnd/base/topics/templateFAQ/Template_rtf.html#escapechar

SAS Institute Inc., *SAS 9.1 Output Delivery System: User's Guide*, Cary, NC: SAS Institute Inc., 2003.

SAS Institute Inc., *SAS Note-008044 TEXT= option on the ODS PRINTER/RTF Statement is Always Left Justified*, <http://support.sas.com/techsup/unotes/SN/008/008044.html>, Feb 2005.

CONTACT INFORMATION

Contact the author at:

Lori S. Parsons
Ovation Research Group,
a division of ICON Clinical Research
Phone: (425) 672-8782
Email: lparsons@ovation.org

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.

APPENDIX I: Complete Code for Example

```

title;
footnote;
options orientation=portrait nodate center pageno=1;
ods escapechar = '^';

ods rtf file = "H:\Admin\SGF07\Output\SGF07.rtf" startpage=no;
/* Define SAS Titles */
/* The title of the first table will be contained in title5 */
title1 j=1 font=elephant height = 30 pt color=dab bcolor=vpab
      "^S={preimage='H:\Admin\SGF07\Logo\Einstein.jpg'}
      Einstein Middle School" ;
title2 j=1 height = 14 pt link="http://www.schools.org/einstein/"
      "http://www.schools.org/einstein/";
title4 j=c font=arial height = 18 pt color=mob bold underlin = 1
      "Summary of Class Enrollment for 2006/2007 ^{super *}";
title6 j=c font=arial height = 14 pt color=bipb bold
      "Gender of Enrolled Students";

/* Turn off output to rtf file */
/* NOTE: This must be done because using the noprint option in PROC FREQ would
   prevent the output file from being created */
ods rtf exclude all;

/* Get data for the first table */
ods select OneWayFreqs;
ods output OneWayFreqs=OneWayFreqs;
proc freq data = sashelp.class;
  table sex;
run;

/* Turn on output to rtf file */
ods rtf exclude none;

/* Print the first table */
proc format;
  value $fsex
    'F' = "Female "
    'M' = "Male ^{super †}";
run;
proc print data = onewayfreqs noobs;
  var sex /
    style(header)=table{font=("Arial, Helvetica",12pt) just=c asis=on
      cellwidth=1.45in}
    style(data)=table{font=("Arial, Helvetica",12pt) just=l asis=on
      cellwidth=1.45in};
  var frequency percent /
    style(header)=table{font=("Arial, Helvetica",12pt) just=c asis=on
      cellwidth=1.45in}
    style(data)=table{font=("Arial, Helvetica",12pt) just=c asis=on
      cellwidth=1.45in};
  format sex $fsex.;
run;

/* Print the custom title for the second table */
/* Code for version 9.1 and higher */
ods rtf text = "^S={outputwidth=100% just=c}
  {\line\line\b\qc\f3\fs28\cf20 Age of Enrolled Students}";

```

Paper 151-2007

```

/* Code for version 8.1 and higher */
/* ods rtf text="{\trowd\trqc\cellx11520\intbl\qc{\b\f3\fs28\cf20
   Age of Enrolled Students\cell}{\row}\pard}"; */

/* Turn off output to rtf file */
ods rtf exclude all;

/* Get data for the second table */
ods select summary;
ods output summary=summary;
proc means data = sashelp.class maxdec=1;
  var age;
run;

/* Turn on output to rtf file */
ods rtf exclude none;

/* Print the second table */
proc print data=summary noobs label;
  var age_mean age_min age_max /
    style(header)=table{font=("Arial, Helvetica",12pt) just=c asis=on
      cellwidth=1.45in}
    style(data)=table{font=("Arial, Helvetica",12pt) just=l asis=on
      cellwidth=1.45in};
  label age_mean = "Mean Age ^{super †}";
  label age_min = "Minimum Age";
  label age_max = "Maximum Age";
run;

/* Print the custom footnotes */
ods rtf text = "^S={outputwidth=100% just=1}
  {\line\line\line _____ \line}";
ods rtf text = "^S={outputwidth=100% just=1}{\f3\fs34\super *}
  {\f3\fs24 Includes summer session.}";
ods rtf text = "^S={outputwidth=100% just=1}{\f3\fs34\super †}
  {\f3\fs24 Two males enrolled in summer session.}";
ods rtf text = "^S={outputwidth=100% just=1}{\f3\fs34\super †}
  {\f3\fs24 Age on first day of class.}";

quit;
ods rtf close;

```